Pensieve header: A Perturbed Alexander Invariant, with formulas from PABI-InBack.nb. Development and experiments notebook.

Programs

In[•]:=
```
Once[<< KnotTheory`];
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

In[•]:=
```
PD[GST48] = PD[X[1, 15, 2, 14], X[29, 2, 30, 3], X[40, 4, 41, 3],
   X[4, 44, 5, 43], X[5, 26, 6, 27], X[95, 7, 96, 6], X[7, 1, 8, 96], X[8, 14, 9, 13],
   X[28, 9, 29, 10], X[41, 11, 42, 10], X[11, 43, 12, 42], X[12, 27, 13, 28],
   X[15, 31, 16, 30], X[61, 16, 62, 17], X[72, 17, 73, 18], X[83, 18, 84, 19],
   X[34, 20, 35, 19], X[20, 89, 21, 90], X[92, 21, 93, 22], X[22, 79, 23, 80],
   X[23, 68, 24, 69], X[24, 57, 25, 58], X[56, 25, 57, 26], X[31, 63, 32, 62],
   X[32, 74, 33, 73], X[33, 85, 34, 84], X[35, 50, 36, 51], X[81, 37, 82, 36],
   X[70, 38, 71, 37], X[59, 39, 60, 38], X[54, 39, 55, 40], X[55, 45, 56, 44],
   X[45, 59, 46, 58], X[46, 70, 47, 69], X[47, 81, 48, 80], X[91, 49, 92, 48],
   X[49, 91, 50, 90], X[82, 52, 83, 51], X[71, 53, 72, 52], X[60, 54, 61, 53],
   X[74, 63, 75, 64], X[85, 64, 86, 65], X[65, 76, 66, 77], X[66, 87, 67, 88],
   X[94, 67, 95, 68], X[86, 75, 87, 76], X[77, 88, 78, 89], X[93, 78, 94, 79]];
```

In[•]:=
```
RVK::usage =
  "RVK[xs, rots] represents a Rotational Virtual Knot with a list of n Xp/Xm crossings
    xs and a length 2n list of rotation numbers rots. Crossing
    sites are indexed 1 through 2n, and rots[[k]] is the rotation
    between site k-1 and site k. RVK is also a casting operator
    converting to the RVK presentation from other knot presentations.";
```

In[•]:=
```
RVK[pd_PD] := Module[{n, xs, x, rots, front = {1}, k},
   n = Length@pd; rots = Table[0, {2 n}];
   xs = Cases[pd, x_X :> {Xp[x[[4]], x[[1]]] PositiveQ@x
                          Xm[x[[2]], x[[1]]]      True   ];
   For[k = 1, k ≤ 2 n, ++k,
    If[FreeQ[front, -k],
     front = Flatten@Replace[front, k → (xs /. {
             Xp[k, l_] | Xm[l_, k] :> {l + 1, k + 1, -l},
             Xp[l_, k] | Xm[k, l_] :> (++rots[[l]]; {-l, k + 1, l + 1}),
             _Xp | _Xm :> {}
            }), {1}],
     Cases[front, k | -k] /. {k, -k} :> --rots[[k]];
    ]
   ];
   RVK[xs, rots] ];
RVK[K_] := RVK[PD[K]];
```

Fast ρ_1



pdf

```
In[•]:= PAI[K_] := Module[{Cs, r, n, B, A, c, s, i, j, Δ, G, g, ρ1},
          {Cs, r} = List @@ RVK[K] /. {Xp[i_, j_] :> {+1, i, j}, Xm[i_, j_] :> {-1, i, j}};
          n = Length[Cs];
          B = Table[0, 2 n, 2 n + 1];
          Do[{s, i, j} = c;
           B[[{i, j}, {i, j, i + 1, j + 1}]] = ({{1, 0, -1, 0}, {0, 1, T^s - 1, -T^s}}), {c, Cs}];
          A = B[[All, 2 ;;]];
          Δ = T^(Total[r] - Total[First /@ Cs])/2 Det[A];
          G = Prepend[Table[0, 2 n]][Inverse[A]]; g_α_,β_ := G[[α, β]];
          ρ1 = Δ^2 Sum[{s, i, j} = c;
              s ((1 - T^s) g_ij (g_ij - g_jj) + 2 g_ii g_ij - g_ij g_ji - g_ii g_jj - g_ij + g_jj - 1 / 2), {c, Cs}];
          ρ1 += Δ^2 Sum[r[[k]] (g_kk - 1 / 2), {k, 2 n}];
          Factor@{Δ, ρ1}];
```

```
In[•]:= PAI[PD[X[4, 2, 5, 1], X[2, 6, 3, 5], X[6, 4, 7, 3]]]
```

Out[•]=

$$\left\{\frac{1-T+T^2}{T},\ -\frac{(-1+T)^2\left(1+T^2\right)}{T^2}\right\}$$

```
In[•]:= -((-1 + T)^2 (1 + T^2))/T^2 // Expand // TeXForm
```

Out[•]//TeXForm=

```
-T^2-\frac{1}{T^2}+2 T+\frac{2}{T}-2
```

pdf

```
In[•]:= Table[Expand@PAI[K], {K, AllKnots[{3, 6}]}] // Column
```

Out[•]=

pdf

$$\left\{-1+\frac{1}{T}+T,\ 2+\frac{1}{T^2}-\frac{2}{T}-2T+T^2\right\}$$
$$\left\{3-\frac{1}{T}-T,\ 0\right\}$$
$$\left\{1+\frac{1}{T^2}-\frac{1}{T}-T+T^2,\ 6+\frac{2}{T^4}-\frac{4}{T^3}+\frac{5}{T^2}-\frac{6}{T}-6T+5T^2-4T^3+2T^4\right\}$$
$$\left\{-3+\frac{2}{T}+2T,\ 18+\frac{5}{T^2}-\frac{14}{T}-14T+5T^2\right\}$$
$$\left\{5-\frac{2}{T}-2T,\ 10+\frac{1}{T^2}-\frac{6}{T}-6T+T^2\right\}$$
$$\left\{-3-\frac{1}{T^2}+\frac{3}{T}+3T-T^2,\ 16+\frac{1}{T^4}-\frac{6}{T^3}+\frac{13}{T^2}-\frac{16}{T}-16T+13T^2-6T^3+T^4\right\}$$
$$\left\{5+\frac{1}{T^2}-\frac{3}{T}-3T+T^2,\ 0\right\}$$

```
In[•]:= Sum[Simplify[Alexander[K][T] == PAI[K][[1]]], {K, AllKnots[{3, 7}]}]
```

Out[•]=

```
14 True
```

```
In[•]:= Timing@Sum[Simplify[Alexander[K][T] == PAI[K][[1]], {K, AllKnots[{3, 10}]}]
```

Out[•]=

$$\{169.094,\ 249\ \text{True}\}$$

```
In[•]:= K = GST48;
        Factor@Alexander[K][T]
```

Out[•]=

$$-\frac{\left(-1+2\,T-T^2-T^3+2\,T^4-T^5+T^8\right)\left(-1+T^3-2\,T^4+T^5+T^6-2\,T^7+T^8\right)}{T^8}$$

pdf

```
In[•]:= Timing@PAI[GST48]
```

Out[•]=
pdf

$$\left\{68.0313,\ \left\{-\frac{\left(-1+2\,T-T^2-T^3+2\,T^4-T^5+T^8\right)\left(-1+T^3-2\,T^4+T^5+T^6-2\,T^7+T^8\right)}{T^8},\right.\right.$$
$$\frac{1}{T^{16}}(-1+T)^2\left(5-18\,T+33\,T^2-32\,T^3+2\,T^4+42\,T^5-62\,T^6-8\,T^7+166\,T^8-242\,T^9+108\,T^{10}+\right.$$
$$\left.\left.\left.132\,T^{11}-226\,T^{12}+148\,T^{13}-11\,T^{14}-36\,T^{15}-11\,T^{16}+148\,T^{17}-226\,T^{18}+132\,T^{19}+108\,T^{20}-\right.\right.\right.$$
$$\left.\left.\left.242\,T^{21}+166\,T^{22}-8\,T^{23}-62\,T^{24}+42\,T^{25}+2\,T^{26}-32\,T^{27}+33\,T^{28}-18\,T^{29}+5\,T^{30}\right)\right\}\right\}$$

```
In[•]:= Alex2[K_] := Module[{Cs, r, n, rot, w, B, A, c, s, i, j, Δ, G, g, a2},
          {Cs, r} = List @@ RVK[K] /. {Xp[i_, j_] :> {+1, i, j}, Xm[i_, j_] :> {-1, i, j}};
          n = Length[Cs]; rot = Total[r]; w = Total[First /@ Cs];
          B = Table[0, 2 n, 2 n + 1];
          Do[{s, i, j} = c;
           B[[{i, j}, {i, i + 1, j, j + 1}]] = {{1, -1, 0, 0}, {0, T^s - 1, 1, -T^s}}, {c, Cs}];
          Δ = T^((rot-w)/2) Det[A = B[[All, 2 ;;]]];
          G = Prepend[Table[0, 2 n]][Inverse[A]]; g_α_,β_ := G[[α, β]];
          a2 = Sum[{s, i, j} = c; s (g_ij - g_jj), {c, Cs}] + (rot + w) / 2;
          Expand@Together@{Δ, T ∂_T Δ, Δ a2}];
        MatrixForm[Alex2 /@ AllKnots[{3, 8}]]
```

Out[•]//MatrixForm=

$$\begin{pmatrix}
-1+\frac{1}{T}+T & -\frac{1}{T}+T & -\frac{1}{T}+T \\
3-\frac{1}{T}-T & \frac{1}{T}-T & \frac{1}{T}-T \\
1+\frac{1}{T^2}-\frac{1}{T}-T+T^2 & -\frac{2}{T^2}+\frac{1}{T}-T+2\,T^2 & -\frac{2}{T^2}+\frac{1}{T}-T+2\,T^2 \\
-3+\frac{2}{T}+2\,T & -\frac{2}{T}+2\,T & -\frac{2}{T}+2\,T \\
5-\frac{2}{T}-2\,T & \frac{2}{T}-2\,T & \frac{2}{T}-2\,T \\
-3-\frac{1}{T^2}+\frac{3}{T}+3\,T-T^2 & \frac{2}{T^2}-\frac{3}{T}+3\,T-2\,T^2 & \frac{2}{T^2}-\frac{3}{T}+3\,T-2\,T^2 \\
5+\frac{1}{T^2}-\frac{3}{T}-3\,T+T^2 & -\frac{2}{T^2}+\frac{3}{T}-3\,T+2\,T^2 & -\frac{2}{T^2}+\frac{3}{T}-3\,T+2\,T^2 \\
-1+\frac{1}{T^3}-\frac{1}{T^2}+\frac{1}{T}+T-T^2+T^3 & -\frac{3}{T^3}+\frac{2}{T^2}-\frac{1}{T}+T-2\,T^2+3\,T^3 & -\frac{3}{T^3}+\frac{2}{T^2}-\frac{1}{T}+T-2\,T^2+3\,T^3 \\
-5+\frac{3}{T}+3\,T & -\frac{3}{T}+3\,T & -\frac{3}{T}+3\,T
\end{pmatrix}$$

$$\begin{pmatrix}
\mathbf{T} & \mathbf{T} & \mathbf{T} \\
3+\frac{2}{T^2}-\frac{3}{T}-3T+2T^2 & -\frac{4}{T^2}+\frac{3}{T}-3T+4T^2 & -\frac{4}{T^2}+\frac{3}{T}-3T+4T^2 \\
-7+\frac{4}{T}+4T & -\frac{4}{T}+4T & -\frac{4}{T}+4T \\
5+\frac{2}{T^2}-\frac{4}{T}-4T+2T^2 & -\frac{4}{T^2}+\frac{4}{T}-4T+4T^2 & -\frac{4}{T^2}+\frac{4}{T}-4T+4T^2 \\
-7-\frac{1}{T^2}+\frac{5}{T}+5T-T^2 & \frac{2}{T^2}-\frac{5}{T}+5T-2T^2 & \frac{2}{T^2}-\frac{5}{T}+5T-2T^2 \\
9+\frac{1}{T^2}-\frac{5}{T}-5T+T^2 & -\frac{2}{T^2}+\frac{5}{T}-5T+2T^2 & -\frac{2}{T^2}+\frac{5}{T}-5T+2T^2 \\
7-\frac{3}{T}-3T & \frac{3}{T}-3T & \frac{3}{T}-3T \\
3-\frac{1}{T^3}+\frac{3}{T^2}-\frac{3}{T}-3T+3T^2-T^3 & \frac{3}{T^3}-\frac{6}{T^2}+\frac{3}{T}-3T+6T^2-3T^3 & \frac{3}{T^3}-\frac{6}{T^2}+\frac{3}{T}-3T+6T^2-3T^3 \\
9-\frac{4}{T}-4T & \frac{4}{T}-4T & \frac{4}{T}-4T \\
-5-\frac{2}{T^2}+\frac{5}{T}+5T-2T^2 & \frac{4}{T^2}-\frac{5}{T}+5T-4T^2 & \frac{4}{T^2}-\frac{5}{T}+5T-4T^2 \\
5-\frac{1}{T^3}+\frac{3}{T^2}-\frac{4}{T}-4T+3T^2-T^3 & \frac{3}{T^3}-\frac{6}{T^2}+\frac{4}{T}-4T+6T^2-3T^3 & \frac{3}{T^3}-\frac{6}{T^2}+\frac{4}{T}-4T+6T^2-3T^3 \\
-7-\frac{2}{T^2}+\frac{6}{T}+6T-2T^2 & \frac{4}{T^2}-\frac{6}{T}+6T-4T^2 & \frac{4}{T^2}-\frac{6}{T}+6T-4T^2 \\
-5+\frac{1}{T^3}-\frac{3}{T^2}+\frac{5}{T}+5T-3T^2+T^3 & -\frac{3}{T^3}+\frac{6}{T^2}-\frac{5}{T}+5T-6T^2+3T^3 & -\frac{3}{T^3}+\frac{6}{T^2}-\frac{5}{T}+5T-6T^2+3T^3 \\
9+\frac{2}{T^2}-\frac{6}{T}-6T+2T^2 & -\frac{4}{T^2}+\frac{6}{T}-6T+4T^2 & -\frac{4}{T^2}+\frac{6}{T}-6T+4T^2 \\
7-\frac{1}{T^3}+\frac{3}{T^2}-\frac{5}{T}-5T+3T^2-T^3 & \frac{3}{T^3}-\frac{6}{T^2}+\frac{5}{T}-5T+6T^2-3T^3 & \frac{3}{T^3}-\frac{6}{T^2}+\frac{5}{T}-5T+6T^2-3T^3 \\
-7+\frac{1}{T^3}-\frac{3}{T^2}+\frac{6}{T}+6T-3T^2+T^3 & -\frac{3}{T^3}+\frac{6}{T^2}-\frac{6}{T}+6T-6T^2+3T^3 & -\frac{3}{T^3}+\frac{6}{T^2}-\frac{6}{T}+6T-6T^2+3T^3 \\
-9-\frac{2}{T^2}+\frac{7}{T}+7T-2T^2 & \frac{4}{T^2}-\frac{7}{T}+7T-4T^2 & \frac{4}{T^2}-\frac{7}{T}+7T-4T^2 \\
13+\frac{1}{T^2}-\frac{7}{T}-7T+T^2 & -\frac{2}{T^2}+\frac{7}{T}-7T+2T^2 & -\frac{2}{T^2}+\frac{7}{T}-7T+2T^2 \\
11+\frac{2}{T^2}-\frac{7}{T}-7T+2T^2 & -\frac{4}{T^2}+\frac{7}{T}-7T+4T^2 & -\frac{4}{T^2}+\frac{7}{T}-7T+4T^2 \\
-11-\frac{2}{T^2}+\frac{8}{T}+8T-2T^2 & \frac{4}{T^2}-\frac{8}{T}+8T-4T^2 & \frac{4}{T^2}-\frac{8}{T}+8T-4T^2 \\
11+\frac{3}{T^2}-\frac{8}{T}-8T+3T^2 & -\frac{6}{T^2}+\frac{8}{T}-8T+6T^2 & -\frac{6}{T^2}+\frac{8}{T}-8T+6T^2 \\
-9+\frac{1}{T^3}-\frac{4}{T^2}+\frac{8}{T}+8T-4T^2+T^3 & -\frac{3}{T^3}+\frac{8}{T^2}-\frac{8}{T}+8T-8T^2+3T^3 & -\frac{3}{T^3}+\frac{8}{T^2}-\frac{8}{T}+8T-8T^2+3T^3 \\
11-\frac{1}{T^3}+\frac{4}{T^2}-\frac{8}{T}-8T+4T^2-T^3 & \frac{3}{T^3}-\frac{8}{T^2}+\frac{8}{T}-8T+8T^2-3T^3 & \frac{3}{T^3}-\frac{8}{T^2}+\frac{8}{T}-8T+8T^2-3T^3 \\
13-\frac{1}{T^3}+\frac{5}{T^2}-\frac{10}{T}-10T+5T^2-T^3 & \frac{3}{T^3}-\frac{10}{T^2}+\frac{10}{T}-10T+10T^2-3T^3 & \frac{3}{T^3}-\frac{10}{T^2}+\frac{10}{T}-10T+10T^2-3T^3 \\
1+\frac{1}{T^3}-\frac{1}{T^2}-T^2+T^3 & -\frac{3}{T^3}+\frac{2}{T^2}-2T^2+3T^3 & -\frac{3}{T^3}+\frac{2}{T^2}-2T^2+3T^3 \\
3+\frac{1}{T^2}-\frac{2}{T}-2T+T^2 & -\frac{2}{T^2}+\frac{2}{T}-2T+2T^2 & -\frac{2}{T^2}+\frac{2}{T}-2T+2T^2 \\
-5-\frac{1}{T^2}+\frac{4}{T}+4T-T^2 & \frac{2}{T^2}-\frac{4}{T}+4T-2T^2 & \frac{2}{T^2}-\frac{4}{T}+4T-2T^2
\end{pmatrix}$$